
Subject: pulldown menu with multiple selections
Posted by [pei zeng](#) on Wed, 27 Oct 1999 07:00:00 GMT
[View Forum Message](#) <> [Reply to Message](#)

Hi, Guys

Does anybody know how to make a pulldown menu with multiple selections?

Thanks!

pei

Subject: Re: pulldown menu with multiple selections
Posted by [davidf](#) on Wed, 27 Oct 1999 07:00:00 GMT
[View Forum Message](#) <> [Reply to Message](#)

R.Bauer (R.Bauer@fz-juelich.de) writes:

> pei zeng wrote:
>
>> Does anybody know how to make a pulldown menu with multiple selections?
>
> I believe this is impossible at the moment.

Oh, nothing is impossible. Not with IDL. :-)

Here is a simple example of a multiple-selection
pull-down menu. Selected items have an asterisk
in front of the item name.

Cheers,

David

```
PRO Example_Button_Events, event
Widget_Control, event.id, Get_Value=buttonValue, Get_UValue=buttonUValue
Widget_Control, event.id, Set_Value=buttonUValue, Set_UValue=buttonValue
END
```

PRO EXAMPLE

```
tlb = Widget_Base(Column=1, Title='Make a Choice...')
selectID = Widget_Button(tlb, Value='Animal Selections...', /Menu, $
    Event_Pro='Example_Button_Events', Scr_XSize=200)
choice1 = Widget_Button(selectID, Value='Choose Dogs', /Menu)
```

```
button = Widget_Button(choice1, Value='Retriever', UValue='* Retriever')
button = Widget_Button(choice1, Value='Boxer', UValue='* Boxer')
button = Widget_Button(choice1, Value='Great Dane', UValue='* Great Dane')
```

```
choice2 = Widget_Button(selectID, Value='Choose Cows', /Menu)
button = Widget_Button(choice2, Value='Holstein', UValue='* Holstein')
button = Widget_Button(choice2, Value='Angus', UValue='* Angus')
button = Widget_Button(choice2, Value='Jersey', UValue='* Jersey')
```

```
Widget_Control, tlb, /Realize
XManager, 'example', tlb, /No_Block
END
```

David Fanning, Ph.D.
Fanning Software Consulting
Phone: 970-221-0438 E-Mail: davidf@dfanning.com
Coyote's Guide to IDL Programming: <http://www.dfanning.com/>
Toll-Free IDL Book Orders: 1-888-461-0155
